



✓ Sjekkliste

- Etterhvert vil monsteret ditt bestå av en mengde figurer og det vil bli nødvendig å sørge for at alle figurene flyttes sammen. I stedet for å legge til flyttkommandoer på alle figurer, kan du rett og slett legge til kontroller på kroppsfiguren og bruke en **send melding** for å kontrollere andre figurer. Når vår tentakkelfigur (eller en hvilken som helst annen figur!) mottar en **flyttet høyre** -melding, kan vi flytte den også til høyre.

```
// på spøkelseskroppen
når [pil høyre v] trykkes
endre x med (hastighet)
send melding [flyttet høyre v]

når [pil venstre v] trykkes
endre x med ((hastighet) * (-1))
send melding [flyttet venstre v]

// på tentakkelen
når jeg mottar [flyttet høyre v]
endre x med (hastighet)

når jeg mottar [flyttet venstre v]
endre x med ((hastighet) * (-1))
```

- Ved å bruke en **send melding** kan vi også endre hvordan bevegelseskommandoen virker, ved å bare endre et skript. **Ta en titt på dette eksempelet:**

```
// på spøkelseskroppen
når [pil høyre v] trykkes
endre x med (hastighet)
pek i retning (90 v)
sprett tilbake ved kanten
hvis <ikke <berører [kant v]?>>
  send melding [flyttet høyre v]
  send melding [flyttet v]
slutt

når [pil venstre v] trykkes
endre x med ((hastighet) * (-1))
pek i retning (-90 v)
sprett tilbake ved kanten
hvis <ikke <berører [kant v]?>>
  send melding [flyttet venstre v]
  send melding [flyttet v]
slutt
```

- Her forteller vi at kroppen skal gå i den retningen den beveger seg i (husk å sette figur orienteringen til **“venstre-høyre”**) og for å stoppe og at den snur hvis den kolliderer med skjermkanten. Kanskje har du også lagt merke til at vi har lagt meldingen i en **hvis** -kloss, vi vil bare at de andre kroppsdelene skal bevege seg hvis kroppen ikke er i kanten av skjermen. **Hvis du bestemmer deg for å legge til animasjon på bevegelsen, må du huske på at tilhørende bevegelser vil ta like lang tid. f. eks hvis å bevege et bein til venstre tar 1 sekund, må alle andre venstrebevegelser ta et sekund også (bruk en **vent** -kloss for å sørge at de er i samme steg).**

- Du kan også animere kroppsfiguren ved å bruke **neste drakt**, som kan aktiveres hver gang en bevegelsesmelding mottas.
- ```
når jeg mottar [flyttet høyre v]
neste drakt
```

**Lisens:** [Code Club World Limited Terms of Service](#) **Forfatter:** Oversatt fra [Code Club UK](#)

**Översetter:** Lars-Erik Wollan